aLIGO Recycling Cavities Optical Layout

Revision History

Version -v1

2/7/2013, D. Coyne: initial release

Notes and Notation

Optica (a TM package for optical design in Mathematica) can be used to ray trace through an existing placement of optics, but it is cumbersome to use for determining the aligned positions of a number of wedged optics (as is the case with the recycling cavity). This notebook determines the positions and orientations of the 3rd recycling mirrors (PR3, SR3), BeamSplitter (BS) and Input Test Masses (ITM) given the following information/assumptions:

1) the X- and Y-arms are assumed to exactly 90 degrees apart. (T960176-C indicates that the deviation from 90 degrees is 1.2 microradian).

2) the x and y axes are aligned parrallel to the Fabry-Perot arm cavity axes, with the origin defined at the intersection of the recycling cavity beams and the splitting surface of the BS
(Note that this is later converted to the LIGO global coordinate system; The LIGO coordinate system has its origin at the projected intersection of the BT cavity axes.)

3) the notation is as follows:
ui =ray unit vector for the beam incident upon surface i
vi = ray unit vector for the beam reflected from surface i
wi = ray unit vector for the beam refracted through surface i
ni = unit normal vector for surface i

i	surface
1	ITMy (HR surface)
2	ITMy (AR surface)
3	CPy (surface adjacent to ITM)
4	CPy (other surface)
5	BS (50/50 surface)
6	PR3 (HR surface)
7	BS (AR surface, y-transmitted ray)
8	SR3 (HR surface)
9	ITMx (HR surface)
10	ITMx (AR surface)

11 CPx (surface adjacent to ITM)
12 CPx (other surface)
13 BS (AR surface, x-transmitted ray)
14 PR2 (HR surface)
15 PRM (HR surface)

ar = anti-reflectance surface
hr = high reflectance surface
bs = beamsplitting surface

d12 = ITMy thickness at center, mm
d23 = ITMy to CPy gap distance, mm
d34 = CPy thickness at center, mm
d45 = CPy to BS(hr) distance, mm
d56 = BS(hr) to PR3 distance, mm
dBS = BS thickness at center, mm
d57 = BS(hr) to BS(ar) for y-beam, mm
d78 = BS(ar), at y-beam intercept, to SR3, mm
d513 = BS(hr) to BS(ar) for x-beam, mm
d910 = ITMx thickness at center, mm
d1011 = ITMx to CPx gap distance, mm
d1112 = CPx thickness at center, mm
d1213 = CPx to BS(ar) distance, mm
d614 = PR3 to PR2 distance, mm
d1415 = PR2 to PRM distance, mm

The “given” parameters (from Zemax layout, etc.) are as follows:
- ITM (hr) center positions from Zemax
- ITM thickness
- ITM to CP gaps
- CP thickness
- BS thickness

■ Intialization

```
Off[General::"spell"]
Off[General::"spell1"]
```

```
Needs["VectorAnalysis`"]
```

```
Needs["Geometry`Rotations`"]
```

Needs[“Geometry`Rotations`”]
should no longer be needed. Use built - in RotationTransform and RotationMatrix

RotationMatrix uses yaw-pitch-roll angles rather than the Euler angles used by RotationMatrix3D
HOWEVER, didn't Minimize function for determination of the BS orientation to work with RotateMatrix3D, so reverted to Rotate3D

```
RotationMatrix3D[phi_,theta_,psi_]:=RotationMatrix[Pi - psi, {0, 0, 1}].RotationMatrix[
  theta, {1, 0, 0}].RotationMatrix[Pi - phi, {0, 0, 1}]
```

```
RotationMatrix3D[phi, theta, psi]
```

```
{{Cos[phi] Cos[psi] - Cos[theta] Sin[phi] Sin[psi], Cos[psi] Sin[phi] + Cos[phi] Cos[theta] Sin[psi], Sin[psi] Sin[theta]},
 {-Cos[psi] Cos[theta] Sin[phi] - Cos[phi] Sin[psi], Cos[phi] Cos[psi] Cos[theta] - Sin[phi] Sin[psi], Cos[psi] Sin[theta]},
 {Sin[phi] Sin[theta], -Cos[phi] Sin[theta], Cos[theta]}}
```

Note: Per J. Wertz, "Spacecraft Attitude Determination and Control", D. Reidel Pub., 1985, pp.763-764, This is a Type 2 Euler Angle representation with a z-x-z rotation sequence.

Needs["Graphics`Shapes`"]
The functionality of `RotateShape`, `TranslateShape`, and `AffineShape` is provided by the newly added kernel functions `Rotate`, `Translate`, `Scale` and `GeometricTransformation`

```
pi = N[π, 10];
```

```
Clear[Reflect];
```

```
Reflect[u_, n_] := Block[{temp, i, j}, temp = CrossProduct[u, n]; i = temp / Sqrt[temp.temp]; j = CrossProduct[n, i]; Return[u.j j - u.n n]];
```

```
Clear[Refract, ni, nt];
Refract[u_, n_, ni_, nt_] :=
  Block[{temp, i, j}, temp = CrossProduct[u, n]; i = temp / Sqrt[temp.temp]; j = CrossProduct[n, i]; temp = ni u.j / nt; Return[temp j - Sqrt[1 - temp^2] n]];
```

Needs["Optica`Optica`"]
SetOptions[DrawSystem, QuickTrace -> False];

```
DMS[rad_] :=
  {IntegerPart[rad 180 / Pi], IntegerPart[FractionalPart[rad 180 / Pi] 60 + Sign[rad] 10^-18], Round[FractionalPart[(rad 180 / Pi) 60] 60]};
```

```
RAD[deg_, min_, sec_] := (deg + min / 60 + sec / 3600) Pi / 180;
```

■ Global to Local Coordinate Transformation Matrices

See Tables 10-14 and Tables 25-27 of T980044-v1(aka -10), "Determination of Local and Global Coordinate Axes for the LIGO Sites".

Rhc -> Hanford Corner station

```
xangle = 619.49 × 10^-6;
yangle = 12.4832 × 10^-6;
```

```
Rhc = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[pi / 2, xangle, -pi / 2];
MatrixForm[Rhc]
```

$$\begin{pmatrix} 1. & 0. & -0.00061949 \\ 7.73322\times 10^{-9} & 1. & 0.0000124832 \\ 0.00061949 & -0.0000124832 & 1. \end{pmatrix}$$

```
MatrixForm[Rhc - IdentityMatrix[3]]
```

$$\begin{pmatrix} -1.91884\times 10^{-7} & 0. & -0.00061949 \\ 7.73322\times 10^{-9} & -7.79151\times 10^{-11} & 0.0000124832 \\ 0.00061949 & -0.0000124832 & -1.91962\times 10^{-7} \end{pmatrix}$$

```
Rhcinv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rhc.Rhcinv]
```

$$\begin{pmatrix} 1. & 1.65436\times 10^{-24} & 1.0842\times 10^{-19} \\ 1.65436\times 10^{-24} & 1. & 0. \\ 1.0842\times 10^{-19} & 0. & 1. \end{pmatrix}$$

```
MatrixForm[Rhcinv]
```

$$\begin{pmatrix} 1. & 7.73322\times 10^{-9} & 0.00061949 \\ 0. & 1. & -0.0000124832 \\ -0.00061949 & 0.0000124832 & 1. \end{pmatrix}$$

Rhxm -> Hanford x-mid station

```
xangle = 305.827 × 10^-6;
yangle = 12.0075 × 10^-6;
```

```
Rhxm = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rhxm]
```

$$\begin{pmatrix} 1. & 0. & -0.000305827 \\ 3.67222\times 10^{-9} & 1. & 0.0000120075 \\ 0.000305827 & -0.0000120075 & 1. \end{pmatrix}$$

```
MatrixForm[Rhxm - IdentityMatrix[3]]
```

$$\begin{pmatrix} -4.67651\times10^{-8} & 0. & -0.000305827 \\ 3.67222\times10^{-9} & -7.209\times10^{-11} & 0.0000120075 \\ 0.000305827 & -0.0000120075 & -4.68372\times10^{-8} \end{pmatrix}$$

```
Rhxminv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rhxm.Rhxminv]
```

$$\begin{pmatrix} 1. & 4.1359\times10^{-25} & 0. \\ 4.1359\times10^{-25} & 1. & 0. \\ 0. & 0. & 1. \end{pmatrix}$$

- **Rhxe -> Hanford x-end station**

```
xangle = -7.8389 10^-6;
yangle = 11.5318 × 10^-6;
```

```
Rhxe = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rhxe]
```

$$\begin{pmatrix} 1. & 0. & 7.8389\times10^{-6} \\ -9.03966\times10^{-11} & 1. & 0.0000115318 \\ -7.8389\times10^{-6} & -0.0000115318 & 1. \end{pmatrix}$$

```
MatrixForm[Rhxe - IdentityMatrix[3]]
```

$$\begin{pmatrix} -3.07242\times10^{-11} & 0. & 7.8389\times10^{-6} \\ -9.03966\times10^{-11} & -6.64913\times10^{-11} & 0.0000115318 \\ -7.8389\times10^{-6} & -0.0000115318 & -9.72155\times10^{-11} \end{pmatrix}$$

```
Rhxeinv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rhxe.Rhxeinv]
```

$$\begin{pmatrix} 1. & 0. & 0. \\ 0. & 1. & 1.69407\times10^{-21} \\ 0. & 1.69407\times10^{-21} & 1. \end{pmatrix}$$

- **Rhym -> Hanford y-mid station**

```
xangle = 619.97 × 10^-6;
yangle = 325.84 × 10^-6;
```

```
Rhym = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rhym]
```

$$\begin{pmatrix} 1. & 0. & -0.00061997 \\ 2.02011\times10^{-7} & 1. & 0.00032584 \\ 0.00061997 & -0.00032584 & 1. \end{pmatrix}$$

```
MatrixForm[Rhym - IdentityMatrix[3]]
```

$$\begin{pmatrix} -1.92181\times10^{-7} & 0. & -0.00061997 \\ 2.02011\times10^{-7} & -5.30859\times10^{-8} & 0.00032584 \\ 0.00061997 & -0.00032584 & -2.45267\times10^{-7} \end{pmatrix}$$

```
Rhyminv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rhym.Rhyminv]
```

$$\begin{pmatrix} 1. & 2.64698\times10^{-23} & 0. \\ 2.64698\times10^{-23} & 1. & 0. \\ 0. & 0. & 1. \end{pmatrix}$$

■ Rhye -> Hanford y-end station

```
xangle = 620.45 × 10^-6;
yangle = 639.20 × 10^-6;
```

```
Rhye = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rhxe]
```

$$\begin{pmatrix} 1. & 0. & 7.8389\times10^{-6} \\ -9.03966\times10^{-11} & 1. & 0.0000115318 \\ -7.8389\times10^{-6} & -0.0000115318 & 1. \end{pmatrix}$$

```
MatrixForm[Rhye - IdentityMatrix[3]]
```

$$\begin{pmatrix} -1.92479\times10^{-7} & 0. & -0.00062045 \\ 3.96592\times10^{-7} & -2.04288\times10^{-7} & 0.0006392 \\ 0.00062045 & -0.0006392 & -3.96767\times10^{-7} \end{pmatrix}$$

```
Rhyeinv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rhye.Rhyeinv]
```

$$\begin{pmatrix} 1. & -5.29396\times10^{-23} & 0. \\ -5.29396\times10^{-23} & 1. & -1.0842\times10^{-19} \\ 0. & -1.0842\times10^{-19} & 1. \end{pmatrix}$$

```
(180 / Pi) 619 × 10^-6 // N
```

0.0354661

▪ Rlc -> Livingston Corner station

```
xangle = 312.0 × 10^-6;
yangle = -611.0 10^-6;
```

```
Rlc = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rlc]
```

$$\begin{pmatrix} 1. & 0. & -0.000312 \\ -1.90632 \times 10^{-7} & 1. & -0.000611 \\ 0.000312 & 0.000611 & 1. \end{pmatrix}$$

```
MatrixForm[Rlc - IdentityMatrix[3]]
```

$$\begin{pmatrix} -4.8672 \times 10^{-8} & 0. & -0.000312 \\ -1.90632 \times 10^{-7} & -1.8666 \times 10^{-7} & -0.000611 \\ 0.000312 & 0.000611 & -2.35332 \times 10^{-7} \end{pmatrix}$$

```
Rlcinv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rlc.Rlcinv]
```

$$\begin{pmatrix} 1. & 0. & 0. \\ 0. & 1. & 1.0842 \times 10^{-19} \\ 0. & 1.0842 \times 10^{-19} & 1. \end{pmatrix}$$

▪ Rlxe -> Livingston x-end station

```
xangle = -315.0 10^-6;
yangle = -610.0 10^-6;
```

```
Rlxe = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rlxe]
```

$$\begin{pmatrix} 1. & 0. & 0.000315 \\ 1.9215 \times 10^{-7} & 1. & -0.00061 \\ -0.000315 & 0.00061 & 1. \end{pmatrix}$$

```
MatrixForm[Rlxe - IdentityMatrix[3]]
```

$$\begin{pmatrix} -4.96125\times10^{-8} & 0. & 0.000315 \\ 1.9215\times10^{-7} & -1.8605\times10^{-7} & -0.00061 \\ -0.000315 & 0.00061 & -2.35662\times10^{-7} \end{pmatrix}$$

```
Rlxeinv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rlxe.Rlxeinv]
```

$$\begin{pmatrix} 1. & 0. & 0. \\ 0. & 1. & 0. \\ 0. & 0. & 1. \end{pmatrix}$$

```
MatrixForm[Rlxeinv - Transpose[Rlxe]]
```

$$\begin{pmatrix} 0. & 0. & 0. \\ 0. & 0. & 0. \\ 0. & 0. & 0. \end{pmatrix}$$

- **Rlye -> Livingston y-end station**

```
xangle = 311.0 × 10^-6;
yangle = 18.8 × 10^-6;
```

```
Rlye = RotationMatrix3D[0, yangle, 0].RotationMatrix3D[Pi / 2, xangle, -Pi / 2];
MatrixForm[Rlye]
```

$$\begin{pmatrix} 1. & 0. & -0.000311 \\ 5.8468\times10^{-9} & 1. & 0.0000188 \\ 0.000311 & -0.0000188 & 1. \end{pmatrix}$$

```
MatrixForm[Rlye - IdentityMatrix[3]]
```

$$\begin{pmatrix} -4.83605\times10^{-8} & 0. & -0.000311 \\ 5.8468\times10^{-9} & -1.7672\times10^{-10} & 0.0000188 \\ 0.000311 & -0.0000188 & -4.85372\times10^{-8} \end{pmatrix}$$

```
Rlyeinv = RotationMatrix3D[Pi / 2, -xangle, -Pi / 2]. RotationMatrix3D[0, -yangle, 0];
MatrixForm[Rlye.Rlyeinv]
```

$$\begin{pmatrix} 1. & 0. & 0. \\ 0. & 1. & 0. \\ 0. & 0. & 1. \end{pmatrix}$$

Vector Analysis for Wedge Position & Orientation Determination

Parameters

wl = 1.064;
nOptic = ModelRefractiveIndex[FusedSilica][WaveLength -> wl]
nAir = ModelRefractiveIndex[Air][WaveLength -> wl]
nVacuum = ModelRefractiveIndex[Vacuum][WaveLength -> wl]

```
nOptic = 1.44963;
nAir = 1.0;
nVacuum = 1.0;
```

Note : D0901920-v12 &-v13 pdf file reports nOptic = 1.44963. In fact Zemax uses Suprasil-Ext index for 20C, zero pressure at 1064 nm which is 1.4500310
According to
- I. H. Malitson. Interspecimen Comparison of the Refractive Index of Fused Silica, J. Opt. Soc. Am. 55, 1205-1208 (1965) doi:10.1364/JOSA.55.001205
- Handbook of Optics, 3rd edition, Vol. 4. McGraw-Hill 2009
as calculated at this URL:
http://refractiveindex.info/wiki/Citing_RefractiveIndex.INFO
the index at 1064 nm is 1.44963

H1

Notes

1) The ITM optic assignments of specific serial numbered optics is given in T1200324-v2. The assignments of the CP and BS optics, and the parameters of each COC optic is given by serial number at https://nebula.ligo.caltech.edu/optics/, with the exception of CPy. GariLynn states that CP09 must be replaced (currently part of H2-ITMy destined to become H1-ITMy) and has chosen CP02 for its replacement.

ITMx is ITM09
ITMy is ITM06
CPx is CP01
CPy is CP02
BS is BS06

2) The magnitude and orientation of the wedge angles are as follows::
ITMx = 0.077 deg, vertical, thick side down
ITMy = 0.078 deg, vertical, thick side down
CPx = 0.069 deg, horizontal, thick side -Y
CPy = 0.069 deg, horizontal, thick side -X

BS = 0.076 deg, horizontal, thick side +X/+Y

Notes:

a) The wedge angle values given in https://nebula.ligo.caltech.edu/optics/ are the "final" values reported by the polishing contractor. However there is some round-off/approx. in this data. GariLynn Billingsley suggests using the values reported in C1107164-v1.

b) The wedge angle magnitudes (but not oreintations/signs) reported in the Zemax optical layout, D0901920-v13 differ, as follows:

ITMx = ITM10? = 0.077 deg (same value but serial number designation is wrong)
ITMy = ITM11? = 0.076 deg (value and serial number incorrect)
CPx = CP01 = 0.070 deg (serial number correct, used "final" wedge angle instead of C1107164-v1 value)
CPy = CP02 = 0.069 deg (serial number is incorrect)
BS = BS06 = 0.076 deg (same)

ITMx = vertical, thick side down
ITMy = vertical, thick side down
CPx = horizontal, thick side - Y
CPy = horizontal, thick side - X
BS = horizontal, thick side + X/+Y

H1 Unique Parameters

```
wedgeITMx = 0.077*pi/180;
wedgeITMy = 0.078*pi/180;
wedgeCPx = 0.069*pi/180;
wedgeCPy = 0.069*pi/180;
wedgeBS = 0.076*pi/180;
```

```
ITMxThick = 200.22;
ITMxDiameter = 340.13;
ITMyThick = 199.64;
ITMyDiameter = 340.06;
CPxThick = 99.82;
CPxDiameter = 340.13;
CPyThick = 99.91;
CPyDiameter = 340.22;
BSThick = 60.41;
BSDiameter = 369.85;
```

```
p1 = {-200, 4983.1, -80};
p9 = {5013, -200, -80};
d12 = ITMyThick - (ITMyDiameter / 2) Tan[wedgeITMy];
d910 = ITMxThick - (ITMxDiameter / 3) Tan[wedgeITMx];
d23 = 20;
d1011 = 20;
d34 = CPyThick - (CPyDiameter / 2) Tan[wedgeCPy];
d1112 = CPxThick - (CPxDiameter / 2) Tan[wedgeCPx];
dBS = BSThick - (BSDiameter / 2) Tan[wedgeBS];
```

Note that the distances d513 and d57 are approximate -- they do not take into account the wedge angle of the BS

■ optic surface orientation determination

■ normal vectors for ITMs and CPs

```
n1 = {0, 1, 0};
n2=RotationMatrix3D[0,wedgeITMy,0].n1
```

```
{0., 0.999999, -0.00136136}
```

```
n3 = n2;
n4 = RotationMatrix3D[0, 0, -wedgeCPy].n3
```

```
{-0.00120428, 0.999998, -0.00136136}
```

```
n9 = {1, 0, 0};
n10 = RotationMatrix3D[pi/2, -wedgeITMx, -(pi/2)].n9
```

```
{0.999999, 0., -0.0013439}
```

```
n11 = n10;
n12 = RotationMatrix3D[0, 0, wedgeCPx].n11
```

```
{0.999998, -0.00120428, -0.0013439}
```

- **ITMy path to BS**

```
u1 = -n1;
w1 = u1;
u2 = w1;
w2 = Refract[u2, n2, nOptic, nVacuum]
```

{0., -1., -0.000612107}

```
u3 = w2;
w3 = Refract[u3, n3, nVacuum, nOptic]
```

$\{0., -1., 2.1684\times10^{-19}\}$

```
u4 = w3;
w4 = Refract[u4, n4, nOptic, nVacuum]
```

{-0.00054148, -1., -0.000612108}

```
u5y = w4;
```

- **ITMx path to BS**

```
u9 = -n9;
w9 = u9;
u10 = w9;
w10 = Refract[u10, n10, nOptic, nVacuum]
```

{-1., 0., -0.00060426}

```
u11 = w10;
w11 = Refract[u11, n11, nVacuum, nOptic]
```

$\{-1., 0., -2.1684\times10^{-19}\}$

```
u12 = w11;
w12 = Refract[u12, n12, nOptic, nVacuum]
```

{-1., -0.00054148, -0.000604261}

```
u13 = w12;
```

Tried to convert from Rotate3D to RotationMatrix3D and didn’t get this to work, so reverting to old function Rotate3D in package “Geometry`Rotations`”
soln = Minimize[Abs[Reflect[u5y, RotationMatrix3D[-yaw, 0, 0].{0, 1, 0}].Refract[Refract[u13, RotatationMatrix3D[wedgeBS, 0, 0].-RotationMatrix3D[yaw, 0, 0].{0, 1, 0}, nVacuum, nOptic], -RotationMatrix3D[yaw, 0, 0].{0, 1, 0}, nOptic, nVacuum]], yaw]

Note also that I have assumed a zero pitch angle. This is approximately correct because the BS has no vertical wedge component. However the reflected and refracted rays have a pitch angle difference of ~8 microrad (very small).

```
Clear[yaw];
soln =
 Maximize[Reflect[u5y, Rotate3D[{0, 1, 0}, -yaw, 0, 0]].Refract[Refract[u13, Rotate3D[-Rotate3D[{0, 1, 0}, -yaw, 0, 0], wedgeBS, 0, 0],
     nVacuum, nOptic], -Rotate3D[{0, 1, 0}, -yaw, 0, 0], nOptic, nVacuum], yaw]
```

```
{1., {yaw → 0.784873}}
```

```
BSyaw = -yaw /. soln[[2]];
BSyawDeg = BSyaw 180 / pi
```

```
-44.9699
```

Compare to Zemax BS yaw angle (45.02885 deg) -- only 22 microradian difference:

```
(90 - 45.02885 + BSyawDeg) pi / 180
```

```
0.0000211772
```

```
n5y = RotationMatrix3D[BSyaw, 0, 0].{0, 1, 0}
n5x = -n5y;
```

```
{-0.706736, 0.707478, 0.}
```

```
v5 = Reflect[u5y, n5y]
```

```
{-1., 0.000507929, -0.000612108}
```

```
w5y = Refract[u5y, n5y, nVacuum, nOptic]
```

```
{0.271849, -0.96234, -0.000422251}
```

```
n13 = RotationMatrix3D[wedgeBS, 0, 0].n5x;
n7 = -n13
```

```
{-0.705797, 0.708415, 0.}
```

```
u7 = w5y;
w7 = Refract[u7, n7, nOptic, nVacuum]
```

```
{0.00050405, -1., -0.000612108}
```

```
w13 = Refract[u13, n13, nVacuum, nOptic];
u5x = w13;
w5x = Refract[u5x, n5x, nOptic, nVacuum]
```

```
{-1., 0.000507926, -0.000604261}
```

BS position determination

```
d513 = dBS / (-w13.n5x)
```

68.9204

```
d57 = dBS / (w5y.n5x)
```

68.9204

```
p2 = p1 + d12 w1
```

{-200., 4783.69, -80.}

```
p3 = p2 + d23 w2
```

{-200., 4763.69, -80.0122}

```
p4 = p3 + d34 w3
```

{-200., 4663.99, -80.0122}

```
p10 = p9 + d910 w9
```

{4812.93, -200., -80.}

```
p11 = p10 + d1011 w10
```

{4792.93, -200., -80.0121}

```
p12 = p11 + d1112 w11
```

{4693.32, -200., -80.0121}

```
Clear[dBSy, dBSx]
soln = Minimize[EuclideanDistance[p4 + dBSy w4, p12 + dBSx w12 + d513 w13], {dBSy, dBSx}]
```

{0.0204444, {dBSy → 4847.8, dBSx → 4829.64}}

```
d45 = dBSy /. soln[[2]];
d1213 = dBSx /. soln[[2]];
```

```
p5 = p4 + d45 w4
p13 = p12 + d1213 w12
p13 + d513 w13
```

{-202.625, -183.81, -82.9796}

{-136.32, -202.615, -82.9304}

{-202.625, -183.81, -82.9592}

Compare to Zemax (H1 : D0901920 - v12) global coordinates for p5 (BS (hr)) = {-202.6, -183.9, -82.9}, so Zemax is correct to within ~0.1 mm

```
SchnuppAssy = (d910 nOptic + d1011 + d1112 nOptic + d1213 + d513 nOptic) - (d12 nOptic + d23 + d34 nOptic + d45)
```

```
82.5752
```

Compare to Zemax (H1 : D0901920 - v12) Schnupp Assymetry of 80 mm, so 2.6 mm too large ... but this is OK.

```
p7 = p5 + d57 u7
```

```
{-183.889, -250.134, -83.0087}
```

PR3 position determination

Take as a given the global x - coordinate of the PR3 from Zemax & IO layout.
pPR3 from Zemax should equal p6

```
pPR3 = {-19741, -174.4, -94.6};
```

```
xPR3 = pPR3[[1]];
```

```
d56 = (xPR3 - p5[[1]]) / v5[[1]]
```

```
19538.4
```

```
p6 = p5 + d56 v5
```

```
{-19741., -173.885, -94.9392}
```

```
p6 - pPR3
```

```
{0., 0.514529, -0.339221}
```

So Zemax position for pPR3 is in error by ~0.5 mm ... which is OK

SR3 position determination

Take as a given the global y - coordinate of the SR3 from Zemax & IO layout.
pSR3 from Zemax should equal p8

```
pSR3 = {-174.6, -19615.9, -94.5};
ySR3 = pSR3[[2]];
```

```
d78 = (ySR3 - p7[[2]]) / w7[[2]]
```

```
19365.8
```

```
p8 = p7 + d78 w7
{-174.128, -19615.9, -94.8627}
p8 - pSR3
{0.472292, 0., -0.362667}
```

So Zemax position for pSR3 is in error by ~0.5 mm ... which is OK

```
n15 = -v14;
```

■ L1

■ Notes

1) The ITM optic assignments of specific serial numbered optics is given in T1200324-v2. The assignments of the CP and BS optics, and the parameters of each COC optic is given by serial number at https://nebula.ligo.caltech.edu/optics/, with the exception of CPy. GariLynn states that CP09 must be replaced (currently part of H2-ITMy destined to become H1-ITMy) and has chosen CP02 for its replacement.

ITMx is ITM04
ITMy is ITM08
CPx is CP06
CPy is CP08
BS is BS02

2) The magnitude and orientation of the wedge angles are as follows::
ITMx = 0.0725 deg, vertical, thick side down
ITMy = 0.074 deg, vertical, thick side down
CPx = 0.073 deg, horizontal, thick side -Y
CPy = 0.066 deg, horizontal, thick side -X
BS = 0.070 deg, horizontal, thick side +X/+Y
Notes:
a) The wedge angle values given in https://nebula.ligo.caltech.edu/optics/ are the "final" values reported by the polishing contractor. However there is some round-off/approx. in this data. GariLynn Billingsley suggests using the values reported in C1107164-v1.
b) The wedge angle magnitudes (but not oreintations/signs) reported in the Zemax optical layout, D0902216-v8 match the serial numbers and wedge angle values given above.

ITMx = vertical, thick side down
ITMy = vertical, thick side down
CPx = horizontal, thick side - Y
CPy = horizontal, thick side - X
BS = horizontal, thick side + X/+Y

L1 Unique Parameters

optic thicknesses are the as-built values reported in C1107164-v1 for the assigned optics

```
wedgeITMx = 0.0725*pi/180;
wedgeITMy = 0.074*pi/180;
wedgeCPx = 0.073*pi/180;
wedgeCPy = 0.066*pi/180;
wedgeBS = 0.070*pi/180;
```

```
ITMxThick = 200.27;
ITMxDiameter = 340.0;
ITMyThick = 199.61;
ITMyDiameter = 339.92;
CPxThick = 100.31;
CPxDiameter = 339.94;
CPyThick = 100.32;
CPyDiameter = 340.11;
BSThick = 59.88;
BSDiameter = 369.98;
```

```
p1 = {-200, 4983.1, -80};
p9 = {5013, -200, -80};
d12 = ITMyThick - (ITMyDiameter / 2) Tan[wedgeITMy];
d910 = ITMxThick - (ITMxDiameter / 3) Tan[wedgeITMx];
d23 = 20;
d1011 = 20;
d34 = CPyThick - (CPyDiameter / 2) Tan[wedgeCPy];
d1112 = CPxThick - (CPxDiameter / 2) Tan[wedgeCPx];
dBS = BSThick - (BSDiameter / 2) Tan[wedgeBS];
```

Note that the distances d513 and d57 are approximate -- they do not take into account the wedge angle of the BS

optic surface orientation determination

normal vectors for ITMs and CPs

```
n1 = {0, 1, 0};
n2=RotationMatrix3D[0,wedgeITMy,0].n1
```

```
{0., 0.999999, -0.00129154}
```

```
n3 = n2;
n4 = RotationMatrix3D[0, 0, -wedgeCPy].n3
```

{-0.00115192, 0.999999, -0.00129154}

```
n9 = {1, 0, 0};
n10 = RotationMatrix3D[pi/2, -wedgeITMx, -(pi/2)].n9
```

{0.999999, 0., -0.00126536}

```
n11 = n10;
n12 = RotationMatrix3D[0, 0, wedgeCPx].n11
```

{0.999998, -0.00127409, -0.00126536}

- **ITMy path to BS**

```
u1 = -n1;
w1 = u1;
u2 = w1;
w2 = Refract[u2, n2, nOptic, nVacuum]
```

{0., -1., -0.000580717}

```
u3 = w2;
w3 = Refract[u3, n3, nVacuum, nOptic]
```

{0., -1., 0.}

```
u4 = w3;
w4 = Refract[u4, n4, nOptic, nVacuum]
```

{-0.000517937, -1., -0.000580718}

```
u5y = w4;
```

- **ITMx path to BS**

```
u9 = -n9;
w9 = u9;
u10 = w9;
w10 = Refract[u10, n10, nOptic, nVacuum]
```

{-1., 0., -0.000568946}

```
u11 = w10;
w11 = Refract[u11, n11, nVacuum, nOptic]
```

```
{-1., 0., 0.}
```

```
u12 = w11;
w12 = Refract[u12, n12, nOptic, nVacuum]
```

```
{-1., -0.00057287, -0.000568947}
```

```
u13 = w12;
```

Tried to convert from Rotate3D to RotationMatrix3D and didn't get this to work, so reverting to old function Rotate3D in package "Geometry`Rotations`"
soln = Minimize[Abs[Reflect[u5y, RotationMatrix3D[-yaw, 0, 0].{0, 1, 0}].Refract[Refract[u13, RotatationMatrix3D[wedgeBS, 0, 0].-RotationMatrix3D[yaw, 0, 0].{0, 1, 0}, nVacuum, nOptic], -RotationMatrix3D[yaw, 0, 0].{0, 1, 0}, nOptic, nVacuum]], yaw]
Note also that I have assumed a zero pitch angle. This is approximately correct because the BS has no vertical wedge component. However the reflected and refracted rays have a pitch angle difference of ~8 microrad (very small).

```
Clear[yaw];
soln =
 Maximize[Reflect[u5y, Rotate3D[{0, 1, 0}, -yaw, 0, 0]].Refract[Refract[u13, Rotate3D[-Rotate3D[{0, 1, 0}, -yaw, 0, 0], wedgeBS, 0, 0],
     nVacuum, nOptic], -Rotate3D[{0, 1, 0}, -yaw, 0, 0], nOptic, nVacuum], yaw]
```

```
{1., {yaw → 0.784942}}
```

```
BSyaw = -yaw /. soln[[2]]
```

```
-0.784942
```

```
BSyawDeg = BSyaw 180 / pi
```

```
-44.9739
```

Compare to Zemax BS yaw angle (45.02885 deg) -- only 22 microradian difference:

```
(90 - 45.02885 + BSyawDeg) pi / 180
```

```
-0.0000477476
```

```
n5y = RotationMatrix3D[BSyaw, 0, 0].{0, 1, 0}
n5x = -n5y;
```

```
{-0.706784, 0.707429, 0.}
```

```
v5 = Reflect[u5y, n5y]
```

```
{-1., 0.000393622, -0.000580718}
```

```
w5y = Refract[u5y, n5y, nVacuum, nOptic]
```

```
{0.271891, -0.962328, -0.000400597}
```

```
n13 = RotationMatrix3D[wedgeBS, 0, 0].n5x;
n7 = -n13
```

```
{-0.70592, 0.708292, 0.}
```

```
u7 = w5y;
w7 = Refract[u7, n7, nOptic, nVacuum]
```

```
{0.000445259, -1., -0.000580718}
```

```
w13 = Refract[u13, n13, nVacuum, nOptic];
u5x = w13;
w5x = Refract[u5x, n5x, nOptic, nVacuum]
```

```
{-1., 0.000393615, -0.000568947}
```

BS position determination

ar = anti-reflectance surface
hr = high reflectance surface
bs = beamsplitting surface

d12 = ITMy thickness at center, mm
d23 = ITMy to CPy gap distance, mm
d34 = CPy thickness at center, mm
d45 = CPy to BS(hr) distance, mm
d56 = BS(hr) to PR3 distance, mm
dBS = BS thickness at center, mm
d57 = BS(hr) to BS(ar) for y-beam, mm
d78 = BS(ar), at y-beam intercept, to SR3, mm
d513 = BS(hr) to BS(ar) for x-beam, mm
d910 = ITMx thickness at center, mm
d1011 = ITMx to CPx gap distance, mm
d1112 = CPx thickness at center, mm
d1213 = CPx to BS(ar) distance, mm

The “given” parameters (from Zemax layout, etc.) are as follows:
 - ITM (hr) center positions from Zemax
 - ITM thickness
 - ITM to CP gaps

- CP thickness
- BS thickness

```
p1 = {-200, 4983.1, -80};
p9 = {5013, -200, -80};
d12 = 199.64 - (340.06 / 2) Tan[wedgeITMy];
d910 = 200.22 - (340.13 / 3) Tan[wedgeITMx];
d23 = 20;
d1011 = 20;
d34 = 99.91 - (340.22 / 2) Tan[wedgeCPy];
d1112 = 99.82 - (340.13 / 2) Tan[wedgeCPx];
dBS = 60.41 - (369.85 / 2) Tan[wedgeBS];
```

Note that the distances d513 and d57 are approximate -- they do not take into account the wedge angle of the BS

```
d513 = dBS / (-w13.n5x)

68.9435

d57 = dBS / (w5y.n5x)

68.9435

p2 = p1 + d12 w1

{-200., 4783.68, -80.}

p3 = p2 + d23 w2

{-200., 4763.68, -80.0116}

p4 = p3 + d34 w3

{-200., 4663.97, -80.0116}

p10 = p9 + d910 w9

{4812.92, -200., -80.}

p11 = p10 + d1011 w10

{4792.92, -200., -80.0114}

p12 = p11 + d1112 w11

{4693.32, -200., -80.0114}
```

```
Clear[dBSy, dBSx]
soln = Minimize[EuclideanDistance[p4 + dBSy w4, p12 + dBSx w12 + d513 w13], {dBSy, dBSx}]
```

```
{0.0407252, {dBSy → 4847.93, dBSx → 4829.5}}
```

```
d45 = dBSy /. soln[[2]];
d1213 = dBSx /. soln[[2]];
```

```
p5 = p4 + d45 w4
p13 = p12 + d1213 w12
p13 + d513 w13
```

```
{-202.511, -183.961, -82.8269}
{-136.182, -202.767, -82.7591}
{-202.511, -183.961, -82.7862}
```

Compare to Zemax (L1 : D0902216 – v8) global coordinates for p5 (BS (hr)) = {–202.5, –184.0, –82.8}, so Zemax is correct to within ~0.1 mm

```
p0 = p5 + n5x BSThick;
```

```
SchnuppAssy = (d910 nOptic + d1011 + d1112 nOptic + d1213 + d513 nOptic) - (d12 nOptic + d23 + d34 nOptic + d45)
```

```
82.3087
```

Compare to Zemax (H1 : D0901920 – v12) Schnupp Assymetry of 80 mm, so 2.6 mm too large ... but this is OK.

```
p7 = p5 + d57 u7
```

```
{-183.766, -250.307, -82.8545}
```

PR3 position determination

Take as a given the global x - coordinate of the PR3 from the IO layout (because the IO layout position is used in the SolidWorks layouts used to physcially place the suspensions
pPR3 from Zemax (D0902216-v8) & IO layout (E1100492-v11, E1100493-v9) should equal p6

```
pPR3zemax = {-19 740.5, -176.3, -93.9};
pPR3io = {-19 740.0, -177.4, -94.5};
xPR3 = pPR3io[[1]];
```

```
d56 = (xPR3 - p5[[1]]) / v5[[1]]
```

```
19 537.5
```

```
p6 = p5 + d56 v5
```

```
{-19 740., -176.271, -94.1727}
```

```
p6 - pPR3zemax
p6 - pPR3io
```

```
{0.5, 0.0293646, -0.272665}
```

```
{0., 1.12936, 0.327335}
```

So Zemax position for pPR3 is in error by ~0.3 mm ... which is OK
IO position for PR3 is in error by ~1.1mm

■ PR2 position determination

PR3 is constrained to be vertical (zero pitch angle) in local coordinates (to minimize vertical bounce mode-to-length coupling). Consequently the normal vector for this optic must be pitched in global coordinates.
Take as a given the global x and y coordinates of the PR2 from the IO layout (because the IO layout position is used in the SolidWorks layouts used to physcially place the suspensions
pPR2 from Zemax (D0902216-v8) & IO layout (E1100492-v11, E1100493-v9) should equal p14

```
pPR2zemax = {-3579.6, -530.4, -93.3};
pPR2io = {-3579.2, -530.4, -94.1};
```

```
u6 = v5;
```

```
Clear[s, yaw]
soln = Minimize[EuclideanDistance[p6 + s Reflect[u6, Rotate3D[Rlc.{1, 0, 0}, -yaw, 0, 0]], pPR2io], {s, yaw}]
```

```
{0.626391, {s → 16 164.7, yaw → -0.0111513}}
```

```
PR3yaw = yaw /. soln[[2]];
d614 = s /. soln[[2]];
n6 = Rotate3D[Rlc.{1, 0, 0}, -PR3yaw, 0, 0]
v6 = Reflect[u6, n6];
p14 = p6 + d614 v6
```

```
{0.999938, -0.0111513, 0.000312}
```

```
{-3579.2, -530.4, -93.4736}
```

```
p14 - pPR2io
p14 - pPR2zemax
```

{-0.000025692, -2.44644 × 10^-6, 0.626391}

{0.399974, -2.44644 × 10^-6, -0.173609}

■ PRM position determination

PR2 is constrained to be vertical (zero pitch angle) in local coordinates (to minimize vertical bounce mode-to-length coupling). Consequently the normal vector for this optic must be pitched in global coordinates.
Take as a given the global x and y coordinates of the PRM from the IO layout (because the IO layout position is used in the SolidWorks layouts used to physcially place the suspensions
pPRM from Zemax (D0902216-v8) & IO layout (E1100492-v11, E1100493-v9) should equal p15

```
pPRMzemax = {-20 190.0, -628.0, -102.8};
pPRMio = {-20 189.6, -628.0, -104.1};
```

```
u14 = v6;
```

```
Clear[s, yaw]
soln = Minimize[EuclideanDistance[p14 + s Reflect[u14, Rotate3D[Rlc.{-1, 0, 0}, -yaw, 0, 0]], pPRMio], {s, yaw}]
```

{0.980667, {s → 16 610.7, yaw → -0.0080166}}

```
PR2yaw = yaw /. soln[[2]];
d1415 = s /. soln[[2]];
n14 = Rotate3D[Rlc.{-1, 0, 0}, -PR2yaw, 0, 0]
v14 = Reflect[u14, n14];
p15 = p14 + d1415 v14
```

{-0.999968, 0.00801671, -0.000312}

{-20 189.6, -628., -103.119}

```
p15 - pPRMio
p15 - pPRMzemax
```

{-0.00057253, 8.42312 × 10^-7, 0.980667}

{0.399427, 8.42312 × 10^-7, -0.319333}

```
u15 = v14;
n15 = -v14;
Clear[yaw]
soln = Maximize[n15.Rotate3D[{1, 0, 0}, -yaw, 0, 0], {yaw}];
PRMyaw = yaw /. soln[[2]]
```

```
0.00587577
```

SR3 position determination

Take as a given the global y - coordinate of the SR3 from Zemax & IO layout.
pSR3 from Zemax should equal p8

```
pSR3 = {-175.2, -19 615.9, -94.1};
ySR3 = pSR3[[2]];
```

```
d78 = (ySR3 - p7[[2]]) / w7[[2]]
```

```
19 365.6
```

```
p8 = p7 + d78 w7
```

```
{-175.143, -19 615.9, -94.1005}
```

```
p8 - pSR3
```

```
{0.0569283, 0., -0.000460114}
```

So Zemax position for pSR3 is in error by ~0.1 mm ... which is OK

SR2 position determination

SR3 is constrained to be vertical (zero pitch angle) in local coordinates (to minimize vertical bounce mode-to-length coupling). Consequently the normal vector for this optic must be pitched in global coordinates.
Take as a given the global x and y coordinates of the SR2 from the Zemax layout (because the Zemax layout positions for the SRC are used in the SolidWorks layouts used to physcially place the suspensions
pSR2 from Zemax (D0902216-v8) should equal p16

```
pSR2zemax = {-594.1, -4178.1, -84.4};
```

```
u8 = w7;
```

```
Clear[s, yaw]
soln = Minimize[EuclideanDistance[p8 + s Reflect[u8, Rotate3D[Rlc.{0, 1, 0}, -yaw, 0, 0]], pSR2zemax], {s, yaw}]
```

```
{0.201491, {s → 15 443.5, yaw → 0.0137885}}
```

```
SR3yaw = yaw /. soln[[2]];
d816 = s /. soln[[2]];
n8 = Rotate3D[Rlc.{0, 1, 0}, -SR3yaw, 0, 0]
v8 = Reflect[u8, n8];
p16 = p8 + d816 v8
```

```
{-0.013788, 0.999905, 0.000611}
```

```
{-594.1, -4178.1, -84.1985}
```

```
p16 - pSR2zemax
```

```
{0.000223243, 0.000194702, 0.201491}
```

■ SRM position determination

SR2 is constrained to be vertical (zero pitch angle) in local coordinates (to minimize vertical bounce mode-to-length coupling). Consequently the normal vector for this optic must be pitched in global coordinates.
Take as a given the global x and y coordinates of the PRM from the Zemax layout (because the Zemax SRC layout positions are used in the SolidWorks layouts used to physcially place the suspensions
pSRM from Zemax (D0902216-v8) should equal p17

```
pSRMzemax = {305.0, -19908.6, -93.2};
```

```
u16 = v8;
```

```
Clear[s, yaw]
soln = Minimize[EuclideanDistance[p16 + s Reflect[u16, Rotate3D[Rlc.{0, -1, 0}, -yaw, 0, 0]], pSRMzemax], {s, yaw}]
```

```
{0.147955, {s → 15756.2, yaw → 0.042113}}
```

```
SR2yaw = yaw /. soln[[2]];
d1617 = s /. soln[[2]];
n16 = Rotate3D[Rlc.{0, -1, 0}, -SR2yaw, 0, 0]
v16 = Reflect[u16, n16];
p17 = p16 + d1617 v16
```

```
{0.0421006, -0.999113, -0.000611}
```

```
{305., -19908.6, -93.348}
```

```
p17 - pSRMzemax
```

$\{-8.93539 \times 10^{-7}, 0.0000358173, -0.147955\}$

```
u17 = v16;
n17 = -v16;
Clear[yaw]
soln = Maximize[n17.Rotate3D[{0, 1, 0}, -yaw, 0, 0], {yaw}];
SRMyaw = yaw /. soln[[2]]
```

```
0.0570943
```

■ PRC, SRC & Schnupp Assymmetry Lengths

According to T0900043-v11, the recycling cavity lengths are supposed to be:
PRC length = 57656.0 mm
SRC length = 56008.0 mm
The Schnupp Assymmetry is supposed to be 80 mm, according to RODA M1200276-v1

```
dPRC = d1415 + d614 + d56 + ((nOptic d12 + d23 + nOptic d34 + d45) + (nOptic d910 + d1011 + nOptic d1112 + d1213 + nOptic d513)) / 2
```

```
57 655.6
```

```
dSRC = d1617 + d816 + d78 + nOptic d57 + ((nOptic d12 + d23 + nOptic d34 + d45) + (nOptic d910 + d1011 + nOptic d1112 + d1213 + nOptic d513)) / 2
```

```
56 007.9
```

```
schnuppAssy = Abs[(nOptic d12 + d23 + nOptic d34 + d45) - (nOptic d910 + d1011 + nOptic d1112 + d1213 + nOptic d513)]
```

```
82.3087
```

■ Summary

```
opticLabel = {"PRM HR", "PR2 HR", "PR3 HR", "BS HR", "BS ARc", "SRM HR", "SR2 HR", "SR3 HR", "CPx", "ITMx", "CPy", "ITMy"};
nOptics = Length[opticLabel];
opticPositionG = {p15, p14, p6, p5, p0, p17, p16, p8, p11, p9, p3, p1};
opticPositionL = opticPositionG.Rlc;
opticNormal = {n15, n14, n6, n5y, n13, n17, n16, n8, n11, n9, n3, n1};
opticYaw = {PRMyaw, PR2yaw, PR3yaw, BSyaw, 0, SRMyaw, SR2yaw, SR3yaw, 0, 0, 0, 0};
```

```
formatedTable := Grid[Flatten[{{
    {"Optic", "Global Coordinates\n(mm)", SpanFromLeft, SpanFromLeft, "Local Coordinates\n(mm)", SpanFromLeft, SpanFromLeft,
     "Normal Unit Vector (global)\n(mm)", SpanFromLeft, SpanFromLeft, "Yaw", "Yaw", SpanFromLeft, SpanFromLeft},
    {SpanFromAbove, "Xg", "Yg", "Zg", "Xl", "Yl", "Zl", "Ug", "Vg", "Wg", "rad", "deg", "min", "sec"}},
   Table[{opticLabel[[i]], NumberForm[opticPositionG[[i, 1]], {10, 1}],
     NumberForm[opticPositionG[[i, 2]], {10, 1}], NumberForm[opticPositionG[[i, 3]], {10, 1}], \
       NumberForm[opticPositionL[[i, 1]], {10, 1}], NumberForm[opticPositionL[[i, 2]], {10, 1}],
     NumberForm[opticPositionL[[i, 3]], {10, 1}], \
       NumberForm[opticNormal[[i, 1]], {10, 6}], NumberForm[opticNormal[[i, 2]], {10, 6}], NumberForm[opticNormal[[i, 3]], {10, 6}], \
       NumberForm[opticYaw[[i]], {10, 6}], NumberForm[DMS[opticYaw[[i]]][[1]], {10, 0}],
     NumberForm[DMS[opticYaw[[i]]][[2]], {10, 0}], NumberForm[DMS[opticYaw[[i]]][[3]], {10, 0}]},
    {i, 1, nOptics}]}, 1],
  Frame →
   All];

formatedTable
```

Optic	Global Coordinates (mm)			Local Coordinates (mm)			Normal Unit Vector (global) (mm)			Yaw	Yaw		
	Xg	Yg	Zg	Xl	Yl	Zl	Ug	Vg	Wg	rad	deg	min	sec
PRM HR	-20189.6	-628.0	-103.1	-20189.6	-628.1	-96.4	0.999983	0.005876	0.000581	0.005876	0.	20.	12.
PR2 HR	-3579.2	-530.4	-93.5	-3579.2	-530.5	-92.0	-0.999968	0.008017	-0.000312	-0.008017	0.	-27.	-34.
PR3 HR	-19740.0	-176.3	-94.2	-19740.0	-176.3	-87.9	0.999938	-0.011151	0.000312	-0.011151	0.	-38.	-20.
BS HR	-202.5	-184.0	-82.8	-202.5	-184.0	-82.7	-0.706784	0.707429	0.000000	-0.784942	-44.	-58.	-26.
BS ARc	-160.2	-226.3	-82.8	-160.2	-226.4	-82.6	0.705920	-0.708292	0.000000	0.000000	0.	0.	0.
SRM HR	305.0	-19908.6	-93.3	305.0	-19908.7	-81.3	-0.057063	0.998370	0.000581	0.057094	3.	16.	17.
SR2 HR	-594.1	-4178.1	-84.2	-594.1	-4178.2	-81.5	0.042101	-0.999113	-0.000611	0.042113	2.	24.	46.
SR3 HR	-175.1	-19615.9	-94.1	-175.2	-19616.0	-82.1	-0.013788	0.999905	0.000611	0.013788	0.	47.	24.
CPx	4792.9	-200.0	-80.0	4792.9	-200.0	-81.4	0.999999	0.000000	-0.001265	0.000000	0.	0.	0.
ITMx	5013.0	-200.0	-80.0	5013.0	-200.0	-81.4	1.000000	0.000000	0.000000	0.000000	0.	0.	0.
CPy	-200.0	4763.7	-80.0	-200.0	4763.6	-82.9	0.000000	0.999999	-0.001292	0.000000	0.	0.	0.
ITMy	-200.0	4983.1	-80.0	-200.0	4983.1	-83.0	0.000000	1.000000	0.000000	0.000000	0.	0.	0.